

Iskalnik mobilnih telefonov

Samo Penič, 64990162

Ljubljana, 1. marec 2006

Kazalo

1	Uvod	3
2	Delovanje vezja	4
2.1	Topologija vezja	4
2.2	Izračun elementov	5
2.3	SPICE Simulacija ojačenja operacijskega ojačevalnika	5
3	Izdelava vezja in težave	7
4	Shema vezja in razporeditev elementov	8
5	Seznam elementov	9
6	Kako naprej?	10
A	Izpis modela lm358.mod	12

1 Uvod

Mobilni telefoni so postali del vsakdanjega življenja. Težko si že predstavljamo, da bi zapustili dom brez telefona v žepu, kaj šele, da bi odšli na potovanje brez njega. Z naraščajočo uporabo mobilnih telefonov pa naraščajo tudi težave povezane z varnostjo, pa naj bodo to tehnične težave zagotavljanja zasebnosti telefonskih pogovorov ali pa čisto družbene omejitve, kot so prepovedi uporabe mobilnih telefonov na določenih območjih. Izdelano vezje bo poskušalo rešiti slednje.

V bankah, na bencinskih črpalkah, bolnišnicah in podobnih javnih prostorih vidimo opozorila, da je uporaba mobilnih telefonov prepovedana, vendar se ne menimo dosti za opozorila in pustimo mobilne telefone kar vključene. Ob vzletu letala smo ravno tako opozorjeni, naj izklopimo mobilne telefone zaradi možnih interferenc z opremo letala. V takšnih situacijah je koristno vedeti, ali je še kdo, ki svojega telefona ni izklopil.

Naše vezje s pomočjo dveh dipolnih anten zaznava elektromagnetno valovanje (imenovano tudi RF – radiofrekvenčno valovanje). Skupaj s tuljavami sta uglaseni na sprejem frekvenc, ki jih uporabljajo mobilni telefoni. Demodulacijo opravlja detektorska dioda in signal, ki ga dioda prepusti ojačamo s pomočjo operacijskega ojačevalnika. Ojačan signal pripeljemo na slušalke. Ob aktivnem delovanju mobilnega telefona slišimo v slušalkah značilne zvoke (ki jih dostikrat tudi slišimo v kakšnem radijskem sprejemniku kot “mo-tnje”). Ob zasuku iskalnika se glasnost v slušalkah povečuje oz. zmanjšuje. Ko sta glasnosti v obeh slušalkah enaki, je iskalnik usmerjen proti izviru RF valovanja.

Naše vezje zaradi postavitve anten omogoča tudi določitev smeri iz katere prihaja signal. To pa močno poveča uporabnost vezja. Zamislite si situacijo, da zjutraj na poti v službo iščete telefon, ki ste ga zvečer nastavili na tiho zvonjenje in potem založili. S pomočjo detektorja mobilnega telefona je situacija hitro rešena.

Idejo za vezje sem dobil v majevi reviji Elektor, letnik 2002[1]. Že od nekdaj so mi bila vseč vezja kot so radijski sprejemniki ali oddajniki, a sem se po nekaj neuspešnih poskusih samogradnje prestrašil vezja, ki imajo opravlja z visokimi frekvencami. Vezje v reviji se mi je zdelo preprosto in upam, da je prvo izmed mnogih RF projektov, ki se jih bom lotil, čeprav sem imel pri izdelavi nemalo težav z izgradnjo pravih tuljav.

Vezje je izdelano po shemi iz revije, razporeditev elementov na ploščici in njihovo oštevilčenje pa je spremenjeno.

2 Delovanje vezja

Vezje je v osnovi razdeljeno na tri dele. Ker so v vezje vključeni aktivni elementi, potrebujemo napajanje. Operacijski ojačevalniki ojačajo nizkofrekvenčni signal, ki ga dobijo iz visokofrekvenčnega dela. V nadaljevanju bo predstavljeno delovanje teh treh enot, vključno z izračunom elementov.

2.1 Topologija vezja

Napajalni del je sestavljen iz LED diode, kondenzatorja C2 in C3 ter upora R5. S pomočjo stikala S1 vezje izklopimo. Vezje napajamo s pomočjo 9V baterije. C2 je gladilni kondenzator, ki med drugim prepreči tudi napajalne konice ob vklopu vezja. Te bi lahko povzročile neprijeten pok v slušalkah vsakič, ko bi vezje vklopili. Zaradi C2 napetost ob vklopu počasneje zraste do napajalne napetosti.

Ker smo uporabili operacijske ojačevalnike z enojnim napajanjem, potrebujemo napetostni nivo, ki idealno leži med 0V in napajalno napetostjo. To bi lahko dosegli z uporovnim delilnikom, sestavljenim z dvema enakima uporoma. Naše vezje pa uporablja LED diodo za dvig napetostnega nivoja nad 0V. Seveda ta napetost ne leži točno na polovici napajalne napetosti, a je dovolj, da lahko operacijski ojačevalnik ojačuje izmenične signale. Ta "referenčni" nivo pripeljemo na neinvertirajoči vhod vsakega od obeh operacijskih ojačevalnikov. Upor R5 omejuje tok skozi LED diodo. Kondenzator C3 predstavlja kratek stik za visokofrekvenčne signale, ki jih ojačevalnik naj ne ojačuje.

Ojačevalnik je sestavljen iz operacijskega ojačevalnika LM358[2]. LM358 je operacijski ojačevalnik z enojnim napajanjem, zato je primeren za baterijsko napajana vezja. V 8 pinskem ohišju se nahajata dva ločena ojačevalnika, ki jih uporabimo za dve simetrični vezji. S pomočjo uporov R1, R2 oziroma R6 in R7 določimo ojačanje operacijskega ojačevalnika, upora R3 in R8 pa sta prisotna za omejitev izhodnega toka in prilagoditev vezja za priklop na nizkoohmsko breme (slušalke). Izhoda operacijskega ojačevalnika sta povezana na slušalke brez blokirnega kondenzatorja, ker deluje ojačevalnik v razredu "A"[1].

Visokofrekvenčni del se začne pri dipolni anteni, ki je sestavljena iz dveh vodnikov določene dolžine. Med oba vodnika je priključena še tuljava, ki z anteno tvori nihajni krog, uglašen na določeno frekvenco. Demodulacijo visokofrekvenčnega signala opravlja dioda, ki s pomočjo upora R4 oz. R9 in

kondenzatorja C1 oz. C4 deluje kot detektor ovojnice moduliranega signala. Anteni sta postavljeni 90° druga na drugo. Tako dobimo dva usmerjena sprejema RF valovanja in lahko z njima določimo smer oddajanja. Ko je glasnost na obeh slušalkah enaka (in največja), smo vezje usmerili točno proti izvoru signala.

2.2 Izračun elementov

Operacijska ojačevalnika sta v orientaciji invertiranega ojačenja. Upora R1 in R2 za prvi oz. R6 in R7 za drugi kanal določata ojačenje po formuli:

$$A_u = -\frac{R1}{R2}$$

Predznak – pove, da je ojačanje negativno. V vezju uporabljene vrednosti sta $R2 = 1k\Omega$ in $R1 = 820k\Omega$, kar nam da ojačanje $A_u = -820 = 58dB$. Zaradi interne frekvenčne kompenzacije operacijskega ojačevalnika bo ojačenje konstantno do frekvence okoli $1kHz$ [2], nakar začne upadati z naklonom $-20dB$ na dekada zaradi interne frekvenčne kompenzacije.

Dipolni anteni imata kraka dolžine $d = \frac{\lambda}{2}$ in sta različna za različne frekvence. Mobilni telefoni delujejo na večjih frekvenčnih pasovih. Od teh naj omenim dva. Najstarejši uporabljen pasje pri $900MHz$, novejši telefoni pa delujejo že pri $1800MHz$ ali višjih. Za $900MHz$ območje sta kraka dipola dolga okoli $7.5cm$ za $1800MHz$ pa $3cm$. Zgradba tuljave ni preveč kritična. Uporabimo lahko lakirano žico od 0.3 do $0.5mm$, ki jo navijemo na tulec premera $5mm$. Število ovojev ki jih naredimo naj bo okoli 10. Če želimo eksperimentirati tudi z nižjimi frekvencami si lahko izdelamo več daljših anten in tuljave z večimi ovoji. S poskušanjem lahko pridemo vse do frekvenc, ki jih oddajajo radijski oddajniki.

Za detektorski diodi uporabimo poljubni germanijevi diodi (recimo AA112, AA16 ali AA119).

2.3 SPICE Simulacija ojačenja operacijskega ojačevalnika

Delovanje operacijskega ojačevalnika sem opazoval tudi s pomočjo modela SPICE. Proizvajalec operacijskega ojačevalnika National semiconductor[3] podaja njegov model kot podvezje, ki ga lahko vključimo v lastno vezje.

V simulaciji sem preveril nastavitve ojačenja operacijskega ojačevalnika v dveh primerih – pri neobremenjenem izhodu in pri obremenjenem izhodu.

Uporabljen je sledeči opis vezja:

Ojacevalnik z LM358

```
.include lm358.mod


Vccp 1 0 9V
Vccn 0 2 0V
Vref 3 0 0V
Vac 6 0 0V AC 0.05V sin(0 0.1 1k 0 0)
R1 4 5 820k
R2 6 4 1k
R3 5 8 220
Rbr 8 0 32
X1 3 4 1 2 5 LM358/NS
.end

.control

ac dec 10 100 100k
plot db(v(8)/v(6))
plot db(v(5)/v(6))

.endc
```

Vsebina lm358.mod je prikazana v dodatku A. Rezultati simulacije, ki jih dobimo ob zagonu programa SpiceOpus[4] so prikazani v grafih na slikah 1 in 2. Iz grafa na sliki 1 vidimo, da se ocenjeno ojačenje pri nizkih frekvencah zelo dobro ujema z izračunom iz prejšnjega razdelka, ravno tako tudi njegov frekvenčni potek, ki smo ga ocenili iz izrisane karakteristike v dokumentaciji operacijskega ojačevalnika. V izbrano breme ne pripeljemo vse moči, ker se je del porabi na upor na izhodu operacijskega ojačevalnika. Tako lahko končno ojačenje na bremenu 32Ω razberemo iz grafa na sliki 2 – $A_{total} = 40dB$.



Slika 1: Ojačenje operacijskega ojačevalnika (v dB) z odprtim izhodom v odvisnosti od frekvence

3 Izdelava vezja in težave

Tiskano vezje ima dimenzije 6.5cm x 6.3cm. Nanj prispajkamo vse elemente. Če želimo uporabiti več različnih anten, je boljše če antene ne prispajkamo neposredno na vezje, ampak pripravimo zanje podnožja.

Vezje bi moralo zaradi svoje enostavnosti delovati brez težav. Težave pa vendarle lahko nastopijo pri visokofrekvenčnem delu. Glasnost signala, ki ga slišimo v slušalkah je zelo odvisna od tipa diode, ki ga uporabimo. Če nimamo pri roki primernih diod, jih lahko vseeno uporabimo s tem, da povečamo ojačenje operacijskega ojačevalnika.

Če kljub pravilni zgradbi ne sprejmemo nobenega signala ali pa je ta zelo šibek, poskusimo spremeniti razmak med ovoji tuljave ali pa dodamo oz. odvzamemo kakšen ovoj.

Pri gradnji so se pojavile še običajne težave pri nakupu elementov. V trgovinah namreč niso imeli odločenih komponent, za katere sem že predvidel, da jih bom vgradil v vezje. Te komponente sem naknadno prispajkal s




Slika 2: Ojačenje operacijskega ojačevalnika (v dB) z obremenjenim izhodom (32Ω) v odvisnosti od frekvence

pomočjo žice na ploščico.

Kljub temeljitemu testiranju različnih, nisem našel optimalnega števila ovojev, zato je signal nekoliko šibkejši od pričakovanega. To pomanjkljivost bi lahko popravil tudi s povečanjem ojačanja operacijskega ojačevalnika.

4 Shema vezja in razporeditev elementov


Na sliki 3 je prikazana vezalna shema. Slika 4 pa prikazuje pogled na ploščico iz strani elementov.



Slika 3: Shema vezja

5 Seznam elementov

Oznaka elementa	Količina	Opis elementa	Vrednost
R1, R6	2	Upor 1/4 W	820kΩ
R2, R7	2	Upor 1/4 W	1kΩ
R3, R8	2	Upor 1/4 W	220Ω
R4, R9	2	Upor 1/4 W	100kΩ
R5	1	Upor 1/4 W	2.2kΩ
C1, C3, C4	3	Kondenzator, večplastni	100nF
C2	1	Kondenzator, elektrolitski 16V	100μF
LED0	1	LED dioda, zelena	
D1, D2	2	Germanijeva dioda	AA116
U1	1	Operacijski ojačevalnik, DIP8	LM358
S1	1	Stikalo ali tipka	
J1	1	Vtičnica za slušalke 3.5mm stereo	
L1, L2	2	Tuljava, 10 ovojev 0.3mm – 0.5mm žice, premer ovojev 0.5mm	
E1, E2, E3, E4	4	Antena, dolžine 7.5cm ali 3cm	



Slika 4: Pogled na ploščico iz zgornje strani

6 Kako naprej?

Vezje do neke mere določi smer, od koder prihaja signal mobilnega telefona, a zaradi raznih medsebojnih sklopov in premajhne usmerjenosti sprejemnih anten ne more določiti točne smeri izvora. Temu prispevajo tudi fiziološki dejavniki, kot sta npr. človeško uho in sluh. Z dodatno kalibracijo bi lahko vezje prilagodili posamezniku in tako vsaj delno izločili ta faktor.

Boljša izboljšava je uporaba komparatorjev in mikrokrmilnika. Tako bi dobili “objektivnejši” detektor smeri. Z dodatki, kot so koračni motor, brezžična ali ožičena komunikacija in možnost postavitve večih iskalnikov mobilnih telefonov na več lokacij bi lahko celo natančno triangulirali izvor signala v prostoru. Takšne izboljšave pa zahtevajo tudi bolj sofisticiran sprejemnik.

Literatura

- [1] *Mobile Phone Sniffer*, B. Kainka. ELEKTOR Electronics, ISSN: 0268-4519, May, 2002.
- [2] *National semiconductor, LM358 Datasheet.*
<http://www.national.com/ds.cgi/LM/LM158.pdf>.
- [3] *National semiconductor, LM358 spice model.*
<http://www.national.com/models/spice/LM/LM358.MOD>.
- [4] *SPICE OPUS Web page.* <http://www.fe.uni-lj.si/spice/welcome.html>.

A Izpis modela lm358.mod

```

*////////////////////////////////////
* (C) National Semiconductor, Inc.
* Models developed and under copyright by:
* National Semiconductor, Inc.

*////////////////////////////////////
* Legal Notice: This material is intended for free software support.
* The file may be copied, and distributed; however, reselling the
* material is illegal

*////////////////////////////////////
* For ordering or technical information on these models, contact:
* National Semiconductor's Customer Response Center
*                               7:00 A.M.--7:00 P.M.  U.S. Central Time
*                               (800) 272-9959
* For Applications support, contact the Internet address:
*  amps-apps@galaxy.nsc.com

*////////////////////////////////////
*LM358 DUAL OPERATIONAL AMPLIFIER MACRO-MODEL
*////////////////////////////////////
*
* connections:      non-inverting input
*                   |   inverting input
*                   |   |   positive power supply
*                   |   |   |   negative power supply
*                   |   |   |   |   output
*                   |   |   |   |   |
*                   |   |   |   |   |
*.SUBCKT LM358/NS    1   2   99   50   28
*
*Features:
*Eliminates need for dual supplies
*Large DC voltage gain =                100dB
*High bandwidth =                      1MHz

```

```

*Low input offset voltage =          2mV
*Wide supply range =          +-1.5V to +-16V
*
*NOTE: Model is for single device only and simulated
*      supply current is 1/2 of total device current.
*      Output crossover distortion with dual supplies
*      is not modeled.
*
*****INPUT STAGE*****
*
IOS 2 1 5N
*^Input offset current
R1 1 3 500K
R2 3 2 500K
I1 99 4 100U
R3 5 50 517
R4 6 50 517
Q1 5 2 4 QX
Q2 6 7 4 QX
*Fp2=1.2 MHz
C4 5 6 128.27P
*
*****COMMON MODE EFFECT*****
*
I2 99 50 75U
*^Quiescent supply current
EOS 7 1 POLY(1) 16 49 2E-3 1
*Input offset voltage.^
R8 99 49 60K
R9 49 50 60K
*
*****OUTPUT VOLTAGE LIMITING*****
V2 99 8 1.63
D1 9 8 DX
D2 10 9 DX
V3 10 50 .635
*
*****SECOND STAGE*****

```

```

*
EH 99 98 99 49 1
G1 98 9 POLY(1) 5 6 0 9.8772E-4 0 .3459
*Fp1=7.86 Hz
R5 98 9 101.2433MEG
C3 98 9 200P
*
*****POLE STAGE*****
*
*Fp=2 MHz
G3 98 15 9 49 1E-6
R12 98 15 1MEG
C5 98 15 7.9577E-14
*
*****COMMON-MODE ZERO STAGE*****
*
*Fpcm=10 KHz
G4 98 16 3 49 5.6234E-8
L2 98 17 15.9M
R13 17 16 1K
*
*****OUTPUT STAGE*****
*
F6 50 99 POLY(1) V6 300U 1
E1 99 23 99 15 1
R16 24 23 17.5
D5 26 24 DX
V6 26 22 .63V
R17 23 25 17.5
D6 25 27 DX
V7 22 27 .63V
V5 22 21 0.27V
D4 21 15 DX
V4 20 22 0.27V
D3 15 20 DX
L3 22 28 500P
RL3 22 28 100K
*

```

```
*****MODELS USED*****  
*  
.MODEL DX D(IS=1E-15)  
.MODEL QX PNP(BF=1.111E3)  
*  
.ENDS  
*$
```